Taylor Polynomials

Mth 351 Summer 2002

June 24, 2002 Maple 6

Bent E. Petersen
Filename: 351u2002_taylor_polys.mws

```
> restart;
```

Maple has a builtin facility for computing Taylor series. In Maple series are a special data type, but we can easily convert them to polynomials. Here is a simple procedure to compute Taylor polynomials:

```
> taylorp:=(fun,cent,deg)->convert(taylor(fun,cent,deg+1),polynom):
```

Here fun is an expression defining a function, cent is an expression of the form z=a which specifies that we want to expand fun in the variable z about the center a, and deg is the degree of the desired Taylor polynomial. Let's check our procedure in a couple of well known cases:

```
> taylorp(sin(x),x=0,11);
```

$$x-\frac{1}{6}x^3+\frac{1}{120}x^5-\frac{1}{5040}x^7+\frac{1}{362880}x^9-\frac{1}{39916800}x^{11}$$

```
> taylorp(cos(x),x=0,10);
```

$$1-\frac{1}{2}x^2+\frac{1}{24}x^4-\frac{1}{720}x^6+\frac{1}{40320}x^8-\frac{1}{3628800}x^{10}$$

To get an idea of how successive Taylor polynomials approximate a function we can use the seq command to form the sequence and then plot each of the polynomials together with the original function. Here is an example (where I have omitted the Taylor polynomial of degree 0 since it is just a constant and not very exciting):

```
> fun01:=exp(sin(t));
```

$$fun01 := \mathbf{e}^{\sin(t)}$$

```
> tlist01:=[fun01, seq(taylorp(fun01,t=0,k), k=1..6)];
```

$$tlist01 := \left[\mathbf{e}^{\sin(t)}, 1+t, 1+t+\frac{1}{2}t^2, 1+t+\frac{1}{2}t^2, 1+t+\frac{1}{2}t^2-\frac{1}{8}t^4, 1+t+\frac{1}{2}t^2-\frac{1}{8}t^4-\frac{1}{15}t^5,\right.$$
$$\left.1+t+\frac{1}{2}t^2-\frac{1}{8}t^4-\frac{1}{15}t^5-\frac{1}{240}t^6\right]$$

We will plot fun01 in blue and the Taylor polynomials in brown:

```
> clist01:=[blue, seq(brown,k=1..6)];
```

$$clist01 := [blue, brown, brown, brown, brown, brown, brown]$$

```
> plot(tlist01,t=-1.4..1.6,color=clist01,thickness=3,title="Some Taylor polynomials");
```



As you can see once we get away from the center (t=0) the Taylor polynomial does not approximate the graph of the original function very well. This behavior is typical for Taylor polynomials.

It may be more illuminating to plot the errors in the Taylor polynomials:

```
> elist01:=[seq(fun01-taylorp(fun01,t=0,k), k=1..6)];
```

$$elist01 := \left[\mathbf{e}^{\sin(t)} - 1 - t, \mathbf{e}^{\sin(t)} - 1 - t - \frac{1}{2} t^2, \mathbf{e}^{\sin(t)} - 1 - t - \frac{1}{2} t^2, \mathbf{e}^{\sin(t)} - 1 - t - \frac{1}{2} t^2 + \frac{1}{8} t^4, \mathbf{e}^{\sin(t)} - 1 - t - \frac{1}{2} t^2 + \frac{1}{8} t^4 + \frac{1}{15} t^5, \mathbf{e}^{\sin(t)} - 1 - t - \frac{1}{2} t^2 + \frac{1}{8} t^4 + \frac{1}{15} t^5 + \frac{1}{240} t^6 \right]$$

```
> ecolors01:=[blue,red,brown,magenta,navy,aquamarine];
```

$$ecolors01 := [blue, red, brown, magenta, navy, aquamarine]$$

```
> plot(elist01,t=-1.2..1.2,color=ecolors01,thickness=3,title="Errors in Taylor polynomials");
```



If you only see 5 polynomials in the plot it is because the Taylor polynomials of degree 2 and of degree 3 coincide (in this example). Let's plot the last 3 of them separately (and also learn how to display the value of a variable in a plot title).

```
> for k from 4 to 6 do
  plot(elist01[k],t=-1.2..1.2,color=ecolors01[k],thickness=3,title=c
  at("Error in Taylor polynomial of degree ",convert(k,string)) );
  od;
>
```

Error in Taylor polynomial of degree 4



Error in Taylor polynomial of degree 5





I hope I have convinced you that Maple is useful for computing Taylor polynomials and for investigating them. Here's a few more examples which would be a pain to do by hand:

```
> fun02:=(x^2+1)*exp(tan(x));
```

$$fun02 := (x^2+1)\,\mathbf{e}^{\tan(x)}$$

```
> taylorp(fun02,x=0,10);
```

$$1+x+\frac{3}{2}x^2+\frac{3}{2}x^3+\frac{7}{8}x^4+\frac{97}{120}x^5+\frac{149}{240}x^6+\frac{359}{720}x^7+\frac{2287}{5760}x^8+\frac{110689}{362880}x^9+\frac{873979}{3628800}x^{10}$$

```
>
> fun03:=(sin(t))^2*cos(t);
```

$$fun03 := \sin(t)^2\cos(t)$$

```
> taylorp(fun03,t=0,10);
```

$$t^2-\frac{5}{6}t^4+\frac{91}{360}t^6-\frac{41}{1008}t^8+\frac{7381}{1814400}t^{10}$$

```
>
> fun04:=log(1+tan(u));
```

$$fun04 := \ln(1+\tan(u))$$

```
> taylorp(fun04,u=Pi/4,5);
```

$$\ln(2)+u-\frac{1}{4}\pi+\frac{1}{2}\left(u-\frac{1}{4}\pi\right)^2+\frac{2}{3}\left(u-\frac{1}{4}\pi\right)^3+\frac{7}{12}\left(u-\frac{1}{4}\pi\right)^4+\frac{2}{3}\left(u-\frac{1}{4}\pi\right)^5$$

[>